

Sage and Linear Algebra Worksheet

FCLA Section MISLE

Robert Beezer

Department of Mathematics and Computer Science

University of Puget Sound

Spring 2021

1 An Invertible Matrix

An invertible 6×6 matrix.

```
A = matrix(QQ, [[ 1,  2, -1, -2, -1, -6],
                 [-2, -3,  1,  3,  1,  6],
                 [ 2,  4, -1, -4,  0, -7],
                 [ 0, -1,  1,  2, -1,  1],
                 [ 2,  4, -2, -4, -1, -8],
                 [ 1,  0,  0,  0, -1,  2]])
```

A

```
Ainv = A.inverse()
Ainv
```

```
Ainv*A
```

```
A*Ainv
```

Building the inverse, now column by column, by solving systems of equations with the vector of constants equal to columns of an identity matrix.

Demonstration 1 Edit and re-run the next cell, using several different columns of the identity matrix. Compare the solution vectors with the columns of the inverse of A above.

```
y = vector(QQ, [1,0,0,0,0,0])
P = A.augment(y, subdivide=True)
P.rref()
```

Consolidating the above as the following illustrates Theorem CINM and its proof.

```
B = A.augment(identity_matrix(6), subdivide=True)
B
```

```
B.rref()
```

2 A Matrix Without an Inverse

Now, a non-invertible 6×6 matrix.

```
C = matrix(QQ, [[-1, 0, 2, 8, 8, 0],
                [-1, -1, 0, 1, 0, 3],
                [-1, -2, -1, -4, -4, 5],
                [ 1, 2, 1, 4, 5, -5],
                [-1, -1, -1, -1, -8, 4],
                [ 0, 1, 0, 3, 5, -1]])
```

```
C.inverse()
```

Demonstration 2 Edit and re-run the next cell, using several different columns of the identity matrix, in attempts to construct columns of the inverse. Just one of these failures is enough to be certain that A does not have an inverse, following the logic of Example MWIAA.

```
y = vector(QQ, [1,0,0,0,0,0])
P = C.augment(y, subdivide=True)
P.rref()
```

It is the case that *no* column of the identity matrix leads to a consistent system (for this matrix). We can see all six systems with a computation reminiscent of Theorem CINM.

```
D = C.augment(identity_matrix(6), subdivide=True)
D
```

```
D.rref()
```

Demonstration 3 You might struggle to find *any* vector of constants that will combine with C to be a consistent systems. (Though there are some. And I can think of seven that are real easy to find if you think about it a bit.)

```
w = vector(QQ, [      ])
P = C.augment(w, subdivide=True)
P.rref()
```

Demonstration 4 Do you see the fundamental difference between the matrices A and B?

This work is Copyright 2016–2019 by Robert A. Beezer. It is licensed under a [Creative Commons Attribution-ShareAlike 4.0 International License](https://creativecommons.org/licenses/by-sa/4.0/).